

# LOWE ELECTRONICS LIMITED

Chesterfield Road, Matlock, Derbyshire. DE4 5LE. Telephone 0629 2817,2430,4057,4995. Telex 377482.

GENIE

HI-RES

INSTRUCTION

MANUAL

(C) Lowe Electronics

April 1982



## HI - RES INSTRUCTIONS

### CONTENTS

1/ INTRODUCTION	2
2/ LOADING INSTRUCTIONS	3
3/ EXTENDED BASIC COMMANDS.	5
4/ TECHNICAL INFORMATION	7
5/ EXAMPLE PROGRAMS	8

## HI - RES INSTRUCTIONS

### 1/ INTRODUCTION

This HI-RES unit has been developed to give you high resolution graphics on your Genie Computer, with a resolution of 384 x 192 ; a total of 73,728 individually addressable pixels.

The HI-RES operates via ten extended Basic commands, such as PLOT, VECT, DISP and HROFF to make operating the HI-RES ( Part number LE18 ) very easy.

The operating software is supplied on cassette, which can be loaded into the computer, occupying about 1.5 K bytes of RAM with, or without an expansion box or disk drive(s).

Alternatively if you have a Genie 1, Genie 2 or our LE19 EPROM board you can purchase the HI-RES software in an EPROM, which will save on memory and loading time, ( Fitting instructions are supplied with the EPROM ) part number EP3.

The HI-RES output can be considered as a separate display section, so when you are writing or debugging your HI-RES picture, you can work with it, without the loss of your normal display simply switching the HI - RES display on and off via the software.

Any ordinary characters or graphic blocks may be reversed when mixed with the HI-RES graphics giving you a reverse video display.

## HI - RES INSTRUCTIONS

### 2/ LOADING INSTRUCTIONS

#### 2.1 Cassette.

To load and execute the software from cassette. First switch on the system and Press New Line to the READY? prompt.

Then type 'SYSTEM' followed by New Line.

The Genie will respond by displaying '\*? '

Rewind the cassette and press the play button. then TYPE 'HIRES' followed by New Line.

When the software has loaded the Genie will again display '\*? '

To execute and initialise the software simply TYPE '/' followed by New Line.

The screen should clear and display the initialization message showing that the system is ready to accept the new HI-RES commands.

#### 2.2 Disk.

To load and execute the software under Disk Basic, first enter into Disk Basic.

Then TYPE 'CMD"T"' followed by New Line, and follow cassette loading instructions in Section 2.1 WITHOUT EXECUTING the program, instead press the 'BREAK' key.

Then TYPE 'CMD"S"' followed by New Line to return to D.O.S.

Using the 'DUMP' instruction ( explained in your D.O.S. manual ) save the HI-RES software with a filename of 'HIRES/CIM'

The start address is 7800H and the end address is 7F01H.

Then to execute the software from disk, via D.O.S. you MUST (from DOS ):-

Type 'LOAD HIRES/CIM' followed by New Line

Then type 'BASIC' followed by New Line

When in Basic type 'SYSTEM' followed by New Line

The Genie will respond with '\*? '

You can then execute the HI-RES software by typing '/30868' followed by New Line.



## HI - RES INSTRUCTIONS

### 2.3 EPROM.

You may have the optional EP3 EPROM fitted in your machine if it is a Genie 1 or Genie 2, if you own a Video Genie System you may use the LE19 EPROM unit externally.

To use the EP3 EPROM first enter into Basic, select the appropriate start address for your needs from those shown below and type:-  
'SYSTEM' followed by New Line.

Following the '\*? ' prompt type '/address' ( where address is one of the following alternatives ) followed by New Line.

The four entry points into the HI-RES EPROM are

1. '12288' - This address enables the use of lower case, and the HI-RES.
2. '12296' - This address enables lower case and keyboard debounce, with the HI-RES.
3. '12302' - This address enables keyboard debounce only with the HI-RES.
4. '12308' - This address only enables the HI-RES software as with the cassette version.

## HI - RES INSTRUCTIONS

### 3/ EXTENDED BASIC COMMANDS.

There are ten extra Basic commands included in the HI-RES software. These can be used in the same way as ordinary Basic instructions, with the exception of HI-RES commands following 'THEN' or 'ELSE', in which case you should precede the HI-RES command with a colon. i.e 'ELSE :REV(P)'.  
The ten Basic commands are:-

3.1 HREN - This enables the HI-RES screen ( switches it on )

3.2 HROFF - This switches off the HI-RES screen.

Commands 3 to 10 use the following mnemonic representation:-

X = 0 to 383 , Pixels across the screen

Y = 0 to 191 , Pixels up the screen

N.B The X , Y coordinates operate as 0 , 0 = bottom left of screen unlike the standard pixel graphics.

P = 0 to 1023

The P value represents the PRINT @ positions Which can be found on page 86 of your green basic manual ( APPENDIX E ). These values begin at the top left of the screen.

B = 0 to 255

The B value represents 1 of a possible 256. Storage positions in the 4 K bytes of scratch pad, used for storing your programmable characters or graphic patterns.

S = 0 to 2

The S value represents the mode in which graphic pixels are set on the screen, 0 = Black , 1 = White , 2 = Xored (reversed).

S1 = 0 to 3

The S1 values 0 to 2 are the same as for 'S' , value 3 duplicates a pattern out of the scratch pad memory and onto the screen ignoring any HI-RES graphics already in the same position.

## HI - RES INSTRUCTIONS

3.3 CLR ( S ) , This command clears the screen to black , white or reversed.

3.4 REV ( P ) , This command reverses a 6 x 9 graphic block over a character on the screen.

3.5 PLOT ( X , Y , S ) , sets one of the 73,728 pixels on the screen at position X,Y in mode S.

3.6 VECT ( X , Y , S ) , draws a line from the-last plotted position to position X,Y in mode S.

3.7 FILL ( X , Y , S ) TO ( X' , Y' , S2 ) , either Fills a block or draws a border. Where S2 = 0 or 1, 0 = fill block and 1 = draw border. X,Y are the start co ordinates of the block and X',Y' the finishing co ordinates.

3.8 HOLD ( P , B ) , Takes a character pattern in a 6x12 block off the screen and Holds it in the scratch pad at the specified position.

3.9 DISP ( P , B , S1 ) , Displays a pattern that as been stored in the scratch pad using the 'HOLD' command.

3.10 VIEW ( X , Y , VAR% ) , This is similar to the POINT command in Basic, exept that it returns VAR% = -1 if the pixel is on or 0 if it is off.

N.B VAR% can be any valid variable but it MUST have either been declared as an integer with the DEFINT statment or by using a '%' sign after the variable to declare it as an integer.

i.e

```
10 DEFINT A           : REM.SET TO INTEGER
20 VIEW ( 100, 50 , A ) : REM TEST IF PIXEL IS ON
30 IF A = -1 THEN PRINT " PIXEL IS ON "
```

OR you could do the same thing by typing

```
10 VIEW ( 100 , 50 , A% )
20 IF A% = -1 THEN PRINT " PIXEL IS ON "
```



## HI - RES INSTRUCTIONS

### 4/ TECHNICAL INFORMATION

The HI-RES unit has its own 16 K bytes of memory in 6 bit words built into it, each bit representing one pixel.

The 16 K bytes of memory is divided into two sections, the first 12 K bytes of memory are used for the screen making up 73,728 pixels, the other 4 K bytes are used for scratch pad memory, for storing any patterns off the screen.

The HI-RES utilises four of the Z-80 ports which are ECH to EFH (236 to 239).

#### 4.1

Port ECH ( 236 ) is used for reading and writing data to the HI-RES screen, and has the following bit pattern:-

Bit 7 = 0 during a frame scan , and 1 during frame blanking. This bit is not normally used, but has been provided so that the user may use a flicker free display by only accessing the screen during screen blanking.

Bit 6 is a mirror image of the on-off port and = 0 if the HI-RES is off and = 1 if the HI-RES is on.

Bits 0 to 5 represent the pixels in a line and are displayed left to right, where a 1 is a white pixel and a 0 a black pixel.

#### 4.2

Port EFH ( 239 ) is used to switch on or off the HI-RES, a 1 will switch it on and a 0 off.

#### 4.3

To set up an address in the HI-RES memory, you need to access two ports, one for the column position and the other for the line position.

Port EDH ( 237 ) sets up the position ( which is latched until re-accessed ) and takes any number from 0 to 63.

Port EEH ( 238 ) sets up the line ( which, is also latched until re-accessed ) and takes a number 0 to 191, which is in the screen portion of the memory and 192 to 255 which is in the scratch pad part of memory.

## HI - RES INSTRUCTIONS

### 5/ EXAMPLE PROGRAMS

#### 5.1/ Example from BASIC using OUT instruction

As an example, if you wish to completely white out the screen with the HI-RES you could use the following BASIC program

```
10 CLS
20 OUT 239 , 1      : REM SWITCH ON THE HI-RES
30 FOR X = 0 TO 63  : REM COLUMN POSITIONS
40 OUT 237 , X      : REM SET UP COLUM.
50 FOR Y = 0 TO 191 : REM ROW PSITION
60 OUT 238 , Y      : REM SET UP ROW
70 OUT 236 , 63     : REM SET ALL PIXELS ON
80 NEXT Y , X       : REM DO FOR ALL POSITIONS
90 END
```

We suggest you study the example programs below showing basic use of the HI-RES unit, these programs can be arranged as useful subroutines if required.

#### 5.2 Fill command

```
10 CLS:CLR(O):HREN
20 FORI=0TO91
30 FILL(I*2,I,2)TO(I*4,I*2,1)
40 NEXT
50 FORI=1TO5000:NEXT
60 CLR(2)
```

This program is essentially the same as the one above, lines 40,50, and 60 have been added which show how easily a simple pattern may be repeated to produce apparently complex displays.

```
10 CLS:CLR(O):HREN
20 FORI=0TO91
30 FILL(I*2,I,2)TO(I*4 I*2,1)
40 FILL(383-I*2,I,2)TO(383-I*4,I*2,1)
50.FILL(I*2,191-I,2)TO(I*4,191-I*2,1).
60 FILL(383-I*2,191-I,2)T0(383-I*4,191-I*2,1)
70 NEXT
80 FORI=1TO5000:NEXT
90 CLR(2)
100 END
```

Try adding 'REM' to the beginning of one two or three of the extra lines (30 to 60), to see how the pattern is developed.

## HI - RES INSTRUCTIONS

### 5.3 Polar plots

```
10 CLR(O) : CLS : HREN: PI=3.1416
100 M=0 : FOR A=0 TO 40*PI STEP .1
110 R=(SIN(A*.675)[2]*.75+.25 : GOSUB 1600 : NEXT
120 END
1000 REM POLAR PLOT SUBROUTINE :: R=RADIUS (0-1) A=ANGLE
1010 XX=192+130*R*COS(A) : YY=96+94*R*SIN(A)
1020 IF M THEN.:VECT (XX,YY,1) ELSE :PLOT (XX,YY,1)
1030 M=1 : RETURN
```

This example produces a polar plot, (think of it. as a '3D' graph). Note the way the ELSE construction is used, (see page 5). Lines 1000-1030 can be used as a subroutine to produce circular or curving lines within your own programs where R=radius and A=angle.

### 5.4 Sphere Plot

```
10 CLS:CLR(0):HREN
20 PI=3.1415927 : XS=150 : YS=90 : XO=192 : YO=96
30 FOR A=PI/2 TO 0 STEP -PI/40
40 L=SIN(A) : FOR B=0 TO PI/2 STEP PI/100
50 Y=SIN(B)*YS : X=L*COS(B)*XS
60 PLOT (XO+X,YO+Y,1) : PLOT (XO-X,YO+Y,1)
70 PLOT (XO+X,YO-Y,1) : PLOT (XO-X,YO-Y,1)
80 NEXT B,A
100 FOR I=1 TO 5000:NEXT I
110 CLR(2)
120 END
```

The program above draws a sphere as a series of great circles rotated around a single axis vertically down the screen.

### 5.5 Example games program

```
10 CLEAR100:DEFINT A-Z:DEFSNG X,Y:CLS:CLR(0):HREN:Z=0
20 FILL(170,0,1)T0(210,3,0)
30 X=190:Y=3:X1=190:Y1=18:I=0:REV(63):OX=X1:Y2=191-7
40 PLOT(X,Y,1):VECT(OX,Y1,0)
50 PLOT(X,Y,1):VECT(X1,Y1,1):OX=X1
60 A$=INKEY$
70 C=C+1:IF C>5 THEN C=0 ELSE 100
80 IF I=0 THEN:REV(63) ELSE:REV(I-1)
90 REV(I):I=I+1:IF I>63 THEN I=0
100 IF A$="" THEN 60
110 A=ASC(A$)
120 IF A=8 THEN X1=X1-1:IF X1<175 THEN X1=175
```

## HI - RES INSTRUCTIONS

```
130 IFA=9THENX1=X1+1:IFX1>205THENX1=205
140 IFA<>32THEN40
150 Y3=Y1-Y:Y4=Y2-Y:X3=X-X1:X2=X-((Y4*X3)/Y3)
160 VIEW(X2,Y2,Z)
170 FORT=1TO2
180 PLOT(X,Y,1):VECT(X2,Y2,2)
190 NEXT:IFZ=0THEN260
200 Q=((I-1)*6)-3:FORT=1TO10
210 FORT1=1TO2:PLOT(Q,191,1):Q1=RND(40)+(Q-20)
:Q2=RND(40)+151
220 IFQ1<0THENQ1=0 ELSE IFQ1>383THENQ1=383
230 VECT(Q1,Q2,2):NEXTT1
240 NEXTT:Q1=Q-21:Q2=Q+21:IFQ1<0THENQ1=0ELSE
IFQ2>383THENQ2=383
250 FILL(Q1,141,0)TO(Q2,191,0):REV(I-1)
260 Z=0
270 GOTO40
```

We suggest you simply try this program, it provides the basis for a popular 'arcade game'. Serious programmers will recognise that use of machine language subroutines will provide a speed increase.